


DICK SMITH

VZ200

Personal Colour Computer



BASIC APPLICATION PROGRAMS

Information in this document is subject to change without notice and does not represent a commitment on the part of the author.

All rights reserved. No part of this book may be reprinted, or reproduced, or utilized in any form or by any electronic, mechanical, or other means, now known or hereafter invented, including photocopying and recording, or in any information storage and retrieval system, without permission in writing from the Publisher.

BASIC PROGRAM EXAMPLES

INTRODUCTION

Your Color Computer is specially designed to serve a wide range of people including beginners with no prior experience in programming knowledge. Therefore, the most friendly and widely used computer language, BASIC, is employed to simplify the learning process. However, the best way for the user to learn programming is to program with the computer. Therefore, this document is specially written to aid the user in understanding some elementary programming techniques, after he is familiar with the BASIC commands given in the color computer BASIC Reference Manual. Please note that these programs are relatively simple, and do not demonstrate the full capability of the Color Computer. The author strongly recommends that the user write his own programs or modify some of the existing programs, once he feels that he has confidence with the Color Computer.

TITLE	PAGE
1. SUM & AVERAGE	3
2. PERMUTATIONS & COMBINATIONS	4
3. HIGHEST COMMON FACTOR (H.C.F.)	5
4. LOWEST COMMON MULTIPLE (L.C.M.)	6
5. PRIME FACTOR	7
6. ROOTS OF QUADRATIC EQUATION	8
7. AREA OF TRIANGLE	9
8. AREA OF POLYGON	10
9. RADIANS & DEGREES	11-12
10. FAHRENHEIT & CELSIUS	13
11. FOOT & METRE	14
12. POUND & KILOGRAM	15
13. GALLON & LITRE	16
14. DEPRECIATION	17
15. SORTING NUMBERS	18
16. SORTING WORDS	19
17. NUMBER GUESSING	20
18. WORD GUESSING	21
19. RANDOM GRAPHICS	22
20. MELODY	23
21. MARK SIX	24

1. SUM & AVERAGE

This program computes the total sum and average of a group of numbers. Can you tell the logic behind the computer?

```
10 REM SUM & AVERAGE
20 CLS
30 PRINT "SUM AND AVERAGE"
40 INPUT "ENTER HOW MANY NOS. "; A
50 FOR I = 1 TO A
60 PRINT "NOS. "; I; "=";
70 INPUT B
80 C=C+B : NEXT
90 PRINT "SUM =" ; C
100 PRINT "AVERAGE =" ; C/A
110 END
```

RUN

```
SUM AND AVERAGE
ENTER HOW MANY NOS.? 5
NOS. 1 =? 10
NOS. 2 =? 20
NOS. 3 =? 30
NOS. 4 =? 40
NOS. 5 =? 50
SUM = 150
AVERAGE = 30
READY
```

2. PERMUTATIONS & COMBINATIONS

Permutations and combinations are 2 popular subjects in modern mathematics. By using this program, you can get the answers quickly. Can you beat the computer in speed and accuracy?

```
10 REM PERMUTATIONS &
20 REM COMBINATIONS
30 CLS
40 PRINT "PERMUTATIONS & ";
50 PRINT "COMBINATIONS"
60 INPUT "ENTER TOTAL NOS. "; A
70 INPUT "ENTER SUBSET NOS. "; B
80 C=1 : D=1
90 IF B>A THEN 30
100 FOR I = A-B+1 TO A
110 IF C*I>1E36 THEN 200
120 C=C*I : NEXT
130 FOR I = 2 TO B
140 D=D*I : NEXT
150 PRINT "PERMUTATIONS =" ; C
160 PRINT "COMBINATIONS =" ; C/D
170 END
200 PRINT "OVERFLOW" : GOTO 60
```

RUN

```
PERMUTATIONS & COMBINATIONS
ENTER TOTAL NOS.? 5
ENTER SUBSET NOS.? 4
PERMUTATIONS = 120
COMBINATIONS = 5
READY
```

3. HIGHEST COMMON FACTOR (H.C.F.)

Just input 2 numbers and this program will tell you the Highest Common Factor.

```
10 REM FIND HCF
20 CLS
30 PRINT "FIND H.C.F."
40 INPUT "ENTER 2 NUMBERS";A,B
50 IF A=0 OR B=0 THEN 100
60 IF A>B THEN A=A-B
70 IF A<B THEN B=B-A
80 IF A<>B THEN 60
90 PRINT "H.C.F. =";A
100 END
```

RUN

```
FIND H.C.F.
ENTER 2 NUMBERS? 20
?? 10
H.C.F. = 10
READY
```

4. LOWEST COMMON MULTIPLE (L.C.M.)

Similar to (H.C.F.) but will give you the Lowest Common Multiple instead of the Highest Common Factor.

```
10 REM FIND LCM
20 CLS
30 PRINT "FIND L.C.M."
40 INPUT "ENTER 2 NUMBERS";A,B
50 IF A=0 OR B=0 THEN 110
60 IF A>B THEN C=A-1 ELSE C=B-1
70 C=C+1
80 IF INT(C/A)<>C/A THEN 70
90 IF INT(C/B)<>C/B THEN 70
100 PRINT "L.C.M. =" ;C
110 END
```

RUN

```
FIND L.C.M.
ENTER 2 NUMBERS? 11
?? 13
L.C.M. = 143
READY
```

5. PRIME FACTOR

This program identifies all the prime factors hidden in any number.

```
10 REM PRIME FACTORS
20 CLS
30 PRINT "PRIME FACTORS"
40 INPUT "ENTER A NUMBER";A
50 IF A=0 THEN 130
60 PRINT SGN(A); : A=ABS(A)
70 FOR I = 2 TO A : B=0
80 IF A/I<>INT(A/I) THEN 100
90 A=A/I : B=B+1 : GOTO 80
100 IF B=0 THEN 120
110 PRINT I;"^";B;
120 NEXT
130 END
```

RUN

```
PRIME FACTORS
ENTER A NUMBER? 240
 1 2 ^ 4 3 ^ 1 5 ^ 1
READY
```

6. ROOTS OF QUADRATIC EQUATION

Generally speaking, Quadratic Equations are in the form of $ax^2+bx+c=0$, where a, b, and c, are the constant coefficients and x is the unknown variable. This program can find out the roots (values of x) for you easily.

```
10 REM ROOTS OF QUADRATIC
20 REM EQUATION
30 CLS
40 PRINT "QUADRATIC EQUATION"
50 PRINT "A*X^2+B*X+C=0"
60 PRINT "ENTER COEFFICIENTS ";
70 PRINT "A,B,C"
80 INPUT A,B,C
90 D=B^2-4*A*C
100 IF D<0 THEN 160
110 D=SQR(D)
120 PRINT "THE ROOTS ARE : "
130 PRINT (-B-D)/(2*A);
140 PRINT (-B+D)/(2*A)
150 GOTO 170
160 PRINT "NO REAL ROOTS"
170 END
```

RUN

```
QUADRATIC EQUATION
A*X^2+B*X+C=0
ENTER COEFFICIENTS A,B,C
? 1
?? 1
?? -12
THE ROOTS ARE :
-4 3
READY
```

7. AREA OF TRIANGLE

The area of a triangle can be determined once the three sides are fixed. Using this program as a guide, can you write a program to find out the area of a circle if I can give you the radius?

```
10 REM AREA OF TRIANGLE
20 CLS
30 PRINT "AREA OF TRIANGLE"
40 PRINT "ENTER 3 SIDES"
50 INPUT A,B,C
60 D=.5*(A+B+C)
70 E=D*(D-A)*(D-B)*(D-C)
80 PRINT "AREA IS";SQR(E)
90 END
```

RUN

```
AREA OF TRIANGLE
ENTER 3 SIDES
? 6
?? 8
?? 10
AREA IS 24
READY
```

8. AREA OF POLYGON

In this program, the area of a regular polygon can be computed. All you have to do is to input the number of sides and its corresponding length.

```
10 REM AREA OF POLYGON
20 CLS : PI=3.1416
30 PRINT "AREA OF REGULAR ";
40 PRINT "POLYGON"
50 INPUT "ENTER NOS. OF SIDES";A
60 INPUT "ENTER LENGTH";B
70 C=PI*(.5*A-1)/A
80 D=A*B*B*TAN(C)/4
90 PRINT "AREA IS";D
100 END
```

RUN

```
AREA OF REGULAR POLYGON
ENTER NOS. OF SIDES? 5
ENTER LENGTH? 4
AREA IS 27.5278
READY
```

9. RADIAN & DEGREE

This program converts any values in radians to degrees, and vice versa.

```
10 REM RADIAN & DEGREE
20 CLS
30 INPUT "FIND RADIAN(1) OR DEGREE(2)";S
40 IF S=1 THEN 140
50 INPUT "RADIAN";B
60 C=B*180/3.1416
70 IF C>360 THEN C=C-360 : GOTO 70
80 PRINT INT(C);"DEGREES"
90 D=(C-INT(C))*60
100 PRINT INT(D);"MINUTES"
110 E=(D-INT(D))*60
120 PRINT INT(E);"SECONDS"
130 END
140 INPUT "DEGREES";A
150 INPUT "MINUTES";B
160 INPUT "SECONDS";C
170 PRINT
180 D=A+B/60+C/3600
190 IF D>360 THEN D=D-360 : GOTO190
200 D=D*3.1416/180
210 PRINT D;"RADIANS"
220 END
```

RUN

```
FIND RADIAN(1) OR DEGREE(2)? 1
DEGREES? 1
MINUTES? 1
SECONDS? 1
```

```
.0177491 RADIANS
READY
```

RUN

FIND RADIAN(1) OR DEGREE(2)? 2

RADIAN? 1

57 DEGREES

17 MINUTES

44 SECONDS

READY

10. FAHRENHEIT & CELSIUS

Similar to Radian & Degree, except this one calculates the conversion in temperature.

```
10 REM DEGREE FAHRENHEIT &
20 REM CELSIUS
30 CLS
40 PRINT "FIND DEGREE-F(1)"
50 INPUT " OR DEGREE-C(2)"; A
60 IF A=2 THEN 110
70 INPUT "DEGREE-C"; B
80 PRINT B; "DEGREE-C =";
90 PRINT B*9/5+32; "DEGREE-F"
100 END
110 INPUT "DEGREE-F"; B
120 PRINT B; "DEGREE-F =";
130 PRINT (B-32)*5/9; "DEGREE-C"
140 END
```

RUN

```
FIND DEGREE-F(1)
  OR DEGREE-C(2)? 1
DEGREE-C? 0
  0 DEGREE-C = 32 DEGREE-F
READY
```

RUN

```
FIND DEGREE-F(1)
  OR DEGREE-C(2)? 2
DEGREE-F? 32
  32 DEGREE-F = 0 DEGREE-C
READY
```

11. FOOT & METRE

Similar to Radian & Degree, except the subjects are Foot & Metre.

```
10 REM FOOT & METRE
20 CLS
30 PRINT "FIND FOOT(1) OR ";
40 INPUT "METRE(2)"; A
50 IF A=1 THEN 100
60 INPUT "FEET"; B
70 PRINT B; "FEET =";
80 PRINT .3048*B; "METRES"
90 END
100 INPUT "METRES"; B
110 PRINT B; "METRES =";
120 PRINT B/.3048; "FEET"
130 END
```

RUN

```
FIND FOOT(1) OR METRE(2)? 1
METRES? 1
  1 METRES = 3.28084 FEET
READY
```

RUN

```
FIND FOOT(1) OR METRE(2)? 2
FEET? 1
  1 FEET = .3048 METRES
READY
```

12. POUND & KILOGRAM

Similar to Radian & Degree, except that Pound & Kilogram are being converted.

```
10 REM POUND & KILOGRAM
20 CLS
30 INPUT "FIND POUND(1) OR KILOGRAM(2)";A
40 IF A=1 THEN 90
50 INPUT "POUNDS";B
60 PRINT B;"POUNDS =";
70 PRINT .4536*B;"KILOGRAMS"
80 END
90 INPUT "KILOGRAMS";B
100 PRINT B;"KILOGRAMS =";
110 PRINT B/.4536;"POUNDS"
120 END
```

RUN

```
FIND POUND(1) OR KILOGRAM(2)? 1
KILOGRAMS? 1
  1 KILOGRAMS = 2.20459 POUNDS
READY
```

RUN

```
FIND POUND(1) OR KILOGRAM(2)? 2
POUNDS? 1
  1 POUNDS = .4536 KILOGRAMS
READY
```

13. GALLON & LITRES

Similar to Radian & Degree, except that Gallons (us) & Litre are used.

```
10 REM GALLON & LITRE
20 CLS
30 INPUT "FIND GALLON(1) OR LITER(2)";A
40 IF A=1 THEN 90
50 INPUT "GALLONS";B
60 PRINT B;"GALLONS =";
70 PRINT 3.785*B;"LITERS"
80 END
90 INPUT "LITERS";B
100 PRINT B;"LITERS =";
110 PRINT B/3.785;"GALLONS"
120 END
```

RUN

```
FIND GALLON(1) OR LITER(2)? 1
LITERS? 1
  1 LITERS = .264201 GALLONS
READY
```

RUN

```
FIND GALLON(1) OR LITER(2)? 2
GALLONS? 1
  1 GALLONS = 3.785 LITERS
READY
```

14. DEPRECIATION

The value of most commodities will decrease after a certain period of time. This program calculates the depreciation value (the difference) once you have input the original price, the depreciation rate and the timing involved.

```
10 REM DEPRECIATION
20 CLS
30 INPUT "ORIGINAL PRICE";A
40 INPUT "DEPRECIATION RATE(%)" ;B
50 INPUT "NO. OF YEARS";C
60 PRINT "DEPRECIATION =";
70 B=B/100
80 D=A*B*(1-B)^(C-1)
90 D=INT(D*10+.5)/10
100 PRINT D : END
```

RUN

```
ORIGINAL PRICE? 1000
DEPRECIATION RATE(%)? 10
NO. OF YEARS? 5
DEPRECIATION = 65.6
READY
```

15. SORTING NUMBERS

If you input a group of numbers (from 2 to 20), this program will sort the numbers in an ascending order. Can you modify the program so that it can sort the numbers in a descending order?

```
10 REM SORTING NOS. IN
20 REM ASCENDING ORDER
30 CLS
40 PRINT "SORTING NOS. (2-20)"
50 INPUT "HOW MANY NOS. "; A
60 DIM A(19)
70 FOR I = 1 TO A
80 PRINT "NO. "; I; : INPUT A(I-1)
90 NEXT
100 FOR J = 0 TO A-2
110 FOR I = 0 TO A-2
120 IF A(I)<A(I+1) THEN 140
130 B=A(I) : A(I)=A(I+1) : A(I+1)=B
140 NEXT : NEXT
150 FOR I = 0 TO A-1
160 PRINT A(I);
170 NEXT
180 END
```

RUN

```
SORTING NOS. (2-20)
HOW MANY NOS.? 6
NO. 1 ? 6
NO. 2 ? 5
NO. 3 ? 4
NO. 4 ? 3
NO. 5 ? 2
NO. 6 ? 1
  1  2  3  4  5  6
READY
```

16. SORTING WORDS

This program sorts a group of words (from 2 to 10) in alphabetic order.

```
10 REM SORTING WORDS IN
20 REM ALPHABETIC ORDER
30 CLS
40 PRINT "SORTING WORDS(2-10)"
50 INPUT "HOW MANY WORDS";A
60 DIM A$(9)
70 FOR I = 1 TO A
80 PRINT "WORD";I; : INPUT A$(I-1)
90 NEXT
100 FOR J = 0 TO A-2
110 FOR I = 0 TO A-2
120 IF A$(I)<A$(I+1) THEN 140
130 B$=A$(I) : A$(I)=A$(I+1) : A$(I+1)=B$
140 NEXT : NEXT
150 FOR I = 0 TO A-1
160 PRINT A$(I);" ";
170 NEXT
180 END
```

RUN

```
SORTING WORDS(2-10)
HOW MANY WORDS? 6
WORD 1 ? ZOO
WORD 2 ? FAST
WORD 3 ? LAZY
WORD 4 ? EAT
WORD 5 ? EAR
WORD 6 ? HELLO
EAR EAT FAST HELLO LAZY ZOO
READY
```

17. NUMBER GUESSING

The computer will generate a number at random (from 1 to 1000) and you will have to guess what is the pre-selected number. How many trials do you need to guess it?

```
10 REM GUESS A NUMBER
20 CLS : C=1
30 A=RND(1000)
40 PRINT "GUESS A NUMBER"
50 INPUT "(1-1000)";B
60 IF B>A THEN PRINT "SMALLER"
70 IF B<A THEN PRINT "LARGER"
80 IF B=A THEN 100
90 C=C+1 : GOTO 40
100 PRINT "YOU ARE RIGHT"
110 PRINT "YOU HAVE TRIED";C;
120 PRINT "TIMES"
130 END
```

RUN

```
GUESS A NUMBER
(1-1000)? 500
SMALLER
GUESS A NUMBER
(1-1000)? 250
LARGER
GUESS A NUMBER
(1-1000)? 300
YOU ARE RIGHT
YOU HAVE TRIED 3 TIMES
READY
```

18. WORD GUESSING

This time you have to guess a 4 letter word. The method of playing is similar to the Number Guessing.

```
10 REM GUESS A WORD
20 CLS
30 C$="FISHRUSHRESTSIDETALKDIRTWORKGIRLJUMPMOOD"
40 I=(RND(10)-1)*4+1
50 A$=MID$(C$,I,4) : S=1
60 PRINT "GUESS A WORD"
70 INPUT "(4 LETTERS)";B$
80 FOR J = 1 TO 4
90 IF MID$(A$,1,J)=MID$(B$,1,J) THEN NEXT
100 PRINT "YOU HAVE";J-1;
110 PRINT "LETTERS RIGHT"
120 IF J<>5 THEN S=S+1 : GOTO 60
130 PRINT "YOU HAVE TRIED";S;
140 PRINT "TIMES"
150 END
```

RUN

```
GUESS A WORD
(4 LETTERS)? R
YOU HAVE 0 LETTERS RIGHT
GUESS A WORD
(4 LETTERS)? S
YOU HAVE 1 LETTERS RIGHT
(4 LETTERS)? SIDE
YOU HAVE 4 LETTERS RIGHT
YOU HAVE TRIED 3 TIMES
READY
```

19. RANDOM GRAPHICS

This simple program produces random patterns on the TV or monitor screen, making use of the pre-defined graphics characters.

```
10 REM GRAPHIC
20 CLS
30 COLOR RND(8)
40 PRINT@ RND(512)-1, "█";
50 GOTO 30
```

20. MELODY

You can write and play your own song. All you have to do is to select the frequency code and the duration code of each note. However, the maximum number of notes that you can play at one time will depend on the memory size of your computer.

```
10 REM SONG
20 CLS
30 INPUT "ENTER NO. OF NOTES";N
40 PRINT "ENTER YOUR NOTES"
50 DIM A%(2*N-1)
60 FOR I = 0 TO N-1
70 INPUT "FREQUENCY CODE";A%(I*2)
80 INPUT "DURATION CODE";A%(I*2+1)
90 NEXT
100 FOR I = 0 TO N-1
110 SOUND A%(I*2),A%(I*2+1)
120 NEXT
```

RUN

```
ENTER NO. OF NOTES? 8
ENTER YOUR NOTES
FREQUENCY CODE? 26
DURATION CODE? 3
FREQUENCY CODE? 30
DURATION CODE? 3
FREQUENCY CODE? 28
DURATION CODE? 3
FREQUENCY CODE? 21
DURATION CODE? 5
FREQUENCY CODE? 26
DURATION CODE? 3
FREQUENCY CODE? 28
DURATION CODE? 3
FREQUENCY CODE? 30
DURATION CODE? 3
FREQUENCY CODE? 26
DURATION CODE? 7
READY
```

21. MARK SIX

This program will generate 6 random numbers with one extra special number.

```
10 REM MARK SIX
20 CLS
30 FOR I = 1 TO 7
40 A(I)=RND(36)
50 IF I = 1 THEN 90
60 FOR J = 1 TO I-1
70 IF A(I)=A(J) THEN 40
80 NEXT
90 NEXT
100 PRINT "THE NOS. ARE : "
110 FOR I = 1 TO 5
120 FOR J = 1 TO 5
130 IF A(J)<A(J+1) THEN 150
140 B=A(J) : A(J)=A(J+1) : A(J+1)=B
150 NEXT : NEXT
160 FOR J = 1 TO 6
170 PRINT A(J);
180 NEXT
190 PRINT
200 PRINT "SPECIAL NO. IS : "
210 PRINT A(7)
220 END
```

RUN

```
THE NOS. ARE :
 4 17 23 30 33 34
SPECIAL NO. IS :
 9
READY
```